Paper SP11-2009

Fitting Compartmental Models to Multiple Dose Pharmacokinetic Data using SAS® PROC NLMIXED

Jing Su, Peng Sun, Xiaodong Li and Alan H. Hartford
Merck & Co., Inc, North Wales, PA

ABSTRACT

Pharmacokinetic (PK) modeling uses systems of ordinary differential equations derived from biological considerations along with statistical models to model the time course of drug in the body. The statistical model requires algorithms for fitting nonlinear mixed effects models. While the NLMIXED procedure in SAS is available, it does not allow for individual subject data to affect the structural form of the model. In the case of a multiple dose study where subjects experience different dosing times, a superposition principle can be used to recursively account for each additional dose.

A SAS template program was written to manipulate data and then construct mean functions for fitting pharmacokinetic data from multiple dose studies. One-compartment models for oral dose administration are considered to illustrate the methodology and challenges for fitting multiple dose data using PROC NLMIXED. The template program contains sample SAS code for fitting two types of models: a general model and a simplified model. The general structural model can handle many situations such as when a subject has irregular dosing intervals, changes dosage during therapy or has non-ignorable differences between actual dosing time and scheduled dosing time, etc. The simplified model is for the most common scenario in which all subjects are constrained to have the same multiple dosing schedules with regular dosing intervals and constant doses. Under each model, we also discuss how to handle missed dose problems.

KEYWORDS: Pharmacokinetic (PK); Compartmental Models; Multiple dose; Superposition; SAS; PROC NLMIXED.

1. INTRODUCTION

Pharmacokinetic (PK) modeling is an area of study where systems of ordinary differential equations (ODEs) are used to model the time course of drug in the body. Statistical models are used in conjunction with the ODEs to account for both intra- and inter-subject variability; PK parameters of the model are estimated for individuals under statistical distribution assumptions and then simulations are performed to determine the effect of different clinical trial or dosing designs.

Specific to PK modeling is that the models do not arise empirically by choosing the simplest algebraic expressions that fit the data. Instead, biological considerations for how the body is known to absorb, distribute, metabolize, and eliminate drug as well as how the mechanism of action of the drug works are used to derive meaningful models to aid in drug development and dosing decisions.

Because of these biological considerations, the functional form of the model, the mean function for a nonlinear statistical model, can be very complex. As the mean function is a component of the solution to an ODE, it might not even have a closed form expression.

Under multiple dose scenarios the superposition principle for accumulating doses is needed to recursively construct the mean function. To illustrate this, assume that a drug is given every 24 hours (i.e., at hours 0, 24, 48 etc.), with the same dose amount at each dosing time. Table 1 and Figure 1 demonstrate the basic idea. Let $C(t)$ denote the drug concentration after a single dose. If $0 \le t < 24$ hrs, then only the first dose is in effect and the predicted concentration is $C(t)$. If $24 \le t < 48$ hrs, then the first two doses are in effect. The contribution of the second dose is $C(t-24)$; hence the predicted concentration is $C(t)+C(t-24)$. The predicted concentrations at other time points can be similarly derived.

Table 1. Recursive Model Formula for Predicting Concentration

Time	Predicted concentration
$0 \le t < 24$ hrs	C(t)
$24 \le t < 48$ hrs	C(t) + C(t-24)
$48 \le t < 72$ hrs	C(t) +C(t-24) +C(t-48)
…	…



Figure 1 Concentration Profiles for Single and Multiple Doses

Further complications are that the form of the mean function varies with different dosing schedules. In real-life situations such as in a clinical trial, the dose may be adjusted based on values of covariates, e.g., weight. Another complication is when one subject misses a dose as compared to all other subjects resulting in this subject requiring a mean function with one fewer peak than the other subjects. Different patients will always have at least slightly different dosing times and differences on the order of only a few minutes may have an impact on parameter estimation.

Because of the distribution assumptions on the parameters, nonlinear mixed model methodology is employed for parameter estimation. However, PROC NLMIXED in SAS is not easy to use when dosing times from subject to subject affect the structural model.

To address the limitation of PROC NLMIXED, a SAS template program was written to manipulate data and then construct mean functions for fitting data from multiple dose studies. One-compartment models for oral dose administration are considered to illustrate the methodology and challenges for fitting multiple dose data using PROC NLMIXED. The template program contains SAS sample codes for fitting two types of models: general model and simplified model. The general model appropriately adjusts the mean function for each subject based on their own realized dosing times. This is needed for cases when a subject changes dosage during therapy or has non-ignorable differences between actual dosing time and scheduled dosing time, etc. The simplified model is for the cases when all subjects are constrained to have exactly the same multiple dosing schedules with regular dosing intervals and constant doses. Unless there are clear discrepancies among dosing schedules, this approach for modeling which ignores small differences between the assigned dosing schedule and the actual dosing times is commonly used in practice for early exploratory modeling. The computational time for the general model can be very slow, especially for large datasets resulting from Phase III trials, but it may be necessary depending on the dosing schedule. Under each model, we also discuss how to handle missed doses.

In Section 2 the basic formulas for single and multiple dose administration are reviewed in the context of one-compartment models with first-order absorption and elimination. Section 3 explains SAS programs for fitting multiple dose pharmacokinetic data. Concluding remarks are in Section 4.

2. ONE-COMPARTMENT MODELS AFTER SINGLE AND MULTIPLE ORAL DOSE ADMINISTRATION

The diagram of a one-compartment pharmacokinetic model with oral administration is displayed in the graph below. The drug first enters the gastrointestinal (GI) tract, then gets absorbed into the central circulation, and is subsequently eliminated from the central circulation.



With first-order absorption and elimination, for a single dose administration with dose amount D administered at time 0, the concentration-time profile $C(t)$ is given by:

$$C(t) = \frac{FDk_a}{(k_a - k_e)V}\left(e^{-k_e t} - e^{-k_a t}\right)$$

where k_a is the absorption rate constant, k_e is the elimination rate constant, F is the bioavailability parameter (fraction of the drug that reaches the systemic circulation), and V is the volume of distribution (the volume in which the amount of drug DF would need to be uniformly distributed to produced the observed concentration). Note that in the context of oral dose administration, only the apparent volume of distribution $V_D = V / F$ is identified. The elimination rate constant can be expressed as $\frac{Cl}{V_D}$, where Cl is the clearance (Cl is interpreted as the volume of blood that is totally clear of drug per unit of time).

For multiple dose administrations, the superposition principle can be applied to derive the concentration-time profile $C(t)$. We start by considering the most general case. Assume that an individual receives m doses at time $t_1, t_2, \ldots, t_m$, respectively. Denote the dose amount at times t_i as D_i ($i = 1,\ldots,m$). To simplify notation, define the dosing intervals as $\tau_i = t_{i+1} - t_i$ for $i = 1,\ldots,m-1$ and let $\tau_0 = t_1$. Appling the superposition principle, we obtain the general model:

$$C(t) = \sum_{i=1}^{n} \frac{D_i k_a}{(k_a - k_e)V_D}\left\{\exp[-k_e(t - \sum_{j=0}^{i-1}\tau_j)] - \exp[-k_a(t - \sum_{l=0}^{i-1}\tau_j)]\right\} \qquad (1)$$

where $n \in \{1,\ldots, m-1\}$ is such that $t_n < t \le t_{n+1}$ and $n = m$ if $t > t_m$,

A special, yet most common case is when $D_i = D$ $(i = 1,\ldots,n)$ and $\tau_i = \tau$ $(i = 1,\ldots,n-1)$. Under this special case, equation (1) can be simplified to:

$$C(t) = \frac{Dk_a}{(k_a - k_e)V_D}\left[\frac{e^{-k_e t}(e^{n\tau k_e} - 1)}{e^{\tau k_e} - 1} - \frac{e^{-k_a t}(e^{n\tau k_a} - 1)}{e^{\tau k_a} - 1}\right] \qquad (2)$$

In clinical trials, it is possible that subjects may miss one or more doses. In this case, the simplified model (2) does not apply but the general model (1) is still applicable by treating missing doses with dose amounts equal to 0. Alternatively, under the special case above, a simplified formula can be derived to account for missed doses and this can significantly reduce the computation time if the number of missed doses is small relative to the total number of doses. If a single dose, say D_{m_1} is missed, then following the simplified model (2), $C(t)$ can be written as

$$C(t) = \frac{Dk_a}{(k_a - k_e)V_D}\left[\frac{e^{-k_e t}(e^{n\tau k_e} - 1)}{e^{\tau k_e} - 1} - \frac{e^{-k_a t}(e^{n\tau k_a} - 1)}{e^{\tau k_a} - 1}\right]$$
$$- \frac{Dk_a}{(k_a - k_e)V_D}\left[e^{-k_e[t-(m_1-1)\tau]} - e^{-k_a[t-(m_1-1)\tau]}\right]\cdot I(m_1 \le n) \qquad (3)$$

where $I(\cdot)$ is the indicator function. $I(m_1 \le n)$ takes a value of 1 if $m_1 \le n$ and 0 otherwise. More generally, if there are k missed doses ($D_{m_1},...,D_{m_k}$), then

$$C(t) = \frac{Dk_a}{(k_a - k_e)V_D}\left[\frac{e^{-k_e t}(e^{n\tau k_e} - 1)}{e^{\tau k_e} - 1} - \frac{e^{-k_a t}(e^{n\tau k_a} - 1)}{e^{\tau k_a} - 1}\right]$$
$$- \frac{Dk_a}{(k_a - k_e)V_D}\left[\sum_{j=1}^{k}\left(e^{-k_e[t-(m_j-1)\tau]} - e^{-k_a[t-(m_j-1)\tau]}\right)\cdot I(m_j \le n)\right] \qquad (4)$$

In the context of nonlinear mixed effects models (Davidian and Giltinan, 1995), pharmacokinetic parameters such as k_a (absorption rate constant), Cl (clearance) and V_D (volume of distribution) are individual specific; but we assume that they come from a common distribution (e.g, a lognormal distribution). In addition, conditional on individual specific pharmacokinetic parameters, the observed drug concentrations are allowed to vary around $C(t)$ so that $C(t)$ is interpreted as the conditional mean or median concentration (depending on the inter-subject error structure). In practice, different inter-subject error structures are fitted to the data including additive, proportional, and exponential (Beal and Sheiner, 1998). Note that while concentration is an observed variable, the random pharmacokinetic parameters, k_a, Cl, and V_D , are not observed; a model is fitted with distributional assumptions for these random parameters and the parameters of these distributions are estimated, e.g., mean and SD of k_a, Cl, and V_D .

In the discussion that follows, we adopt parameterization of k_a, Cl, and V_D with independent lognormal distributions for inter-individual variability and an additive error structure for within-subject variability. Let k_{a_i}, Cl_i, V_{D_i} be subject i's $(i = 1,2,...,m)$ absorption rate constant, clearance, apparent volume of distribution respectively. For each individual i, we use y_{ij} and C_{ij} $(j = 1,...,n_i)$ to denote the observed concentration and conditional mean concentration (as a function of individual specific pharmacokinetic parameters) at time t_{ij} respectively. Then

$$y_{ij} = C_{ij} + \varepsilon_{ij},\ \varepsilon_{ij} \sim i.i.d\ N(0,\sigma^2),$$
$$k_{a_i} = \theta_1 \exp(\eta_{1i}),$$
$$Cl_i = \theta_2 \exp(\eta_{2i}),$$
$$V_{D_i} = \theta_1 \exp(\eta_{3i}),$$

where $(\eta_{1i}, \eta_{2i}, \eta_{3i})$ are assumed to be independent across individuals and follow a multivariate normal distribution with mean (0, 0, 0) and covariance matrix Σ. The SAS programs discussed below can be readily adapted to accommodate situations such as alternative parameterization, different error structures, or more complex structural models.

3. SAS PROGRAMS FOR FITTING MULTIPLE DOSES PK DATA

In this section, we describe the template SAS program for fitting the general and simplified models described in Section 2. To fit each model, we discuss in detail about its required analysis data structure, how to construct the mean function, and how to handle missed doses.

TO FIT THE GENERAL MODEL

The general model is applicable to many situations such as when a subject changes dosage during therapy or has non-ignorable differences between actual dosing time and scheduled dosing time, etc. For simplicity, we assume that in a multiple dose study each individual took three 100 mg oral doses, at 0, 24 and 48 hours. We further assume that PK samples were taken at hours 1, 4, 12, 24, 48, 49, 52, 60, and 72. To construct the mean function and allow for the most flexibility, we propose the following analysis data structure (illustrated for one individual):

Table 2. Data Structure to Fit the General Model (1)

ID	n	time	y	dose_1	dose_2	dose_3	time_1	time_2	time_3
1	1	1	2.7878	100	0	0	1	0	0
1	1	4	4.8740	100	0	0	4	0	0
1	1	12	4.0114	100	0	0	12	0	0
1	1	24	3.3118	100	0	0	24	0	0
1	2	48	4.8601	100	100	0	48	24	0
1	3	49	7.5029	100	100	100	49	25	1
1	3	52	9.2563	100	100	100	52	28	4
1	3	60	7.8583	100	100	100	60	36	12
1	3	72	5.7894	100	100	100	72	48	24

In the above table, *ID* is the individual identification variable, *n* is the number of doses in effect, and *time* is the time of observation. Since the concentration sampled at time 24 was taken right before the 2nd dose, the number of doses in effect was still one and then *n* is 1 for that record. Corresponding to each dose *i*, we create two variables: *dose_i* and *time_i*. If at an observation time, the i th dose is in effect, then *dose_i* is the dose amount of the i th dose and *time_i* is the observation time relative to the i th dosing time; if the i th dose is not in effect, then both *dose_i* and *time_i* take the value of 0. For example, at hour 4, only the first dose is in effect, hence only *dose_1* and *time_1* take non-zero positive values.

The sample SAS code for creating these time variables is:

```
retain time_1 - time_3 0;
array t[3] time_1-time_3;
do i=1 to n;
   t(i)=time-(i-1)*24;
end;
```

As illustrated in Table 1, the mean function is different after each dose. Instead of using if/then statements to specify the exact mean function after each dose, the general model (1) can be directly applied with this special data structure since the variables *dose_i* and *time_i* are set to 0 for the records before the dose *i* was given.

To use the NLMIXED procedure to estimate the parameters k_a, Cl, and V, and their variances as discussed in Section 2, reasonable starting values for these parameters are very important. Some approaches for obtaining starting values might include:

1. Results from the first in man (FIM) study, or domain knowledge.
2. The meaning of the parameters. For example, k_a is the absorption rate constant, and should be between 0 and 1.

Starting value of variance for each parameter can be obtained using the NLIN procedure. Based on the results of fitting a nonlinear model for each subject separately using PROC NLIN, the sample variance of each estimated parameter can be calculated. To make it easier to pass those starting values of variances into the PROC NLMIXED, macro variables can be created.

Let pkdata.sas7bdat be a dataset having the same data structure as presented above. Assume that the starting values for V, k_a, and Cl are 20, 0.8, and 0.6, respectively. Below is the sample code for obtaining the starting values of variances using PRON NLIN:

```
/****to get starting values for variance using PROC NLIN****/
proc nlin data=pkdata;
     by id;
parms v=20
      ka=0.8
      cl=0.6
      ;
k=cl/v;

cst=ka/(v*(ka-k));

model y = cst*dose_1*(exp(-k*time_1)-exp(-ka*time_1))
        + cst*dose_2*(exp(-k*time_2)-exp(-ka*time_2))
        + cst*dose_3*(exp(-k*time_3)-exp(-ka*time_3));

ods output parameterEstimates = estimate ANOVA=ANOVA;
run;
```

Using the starting values of those parameters and their variances and following the same notations as in Section 2, the sample code for fitting the general model (1) using PROC NLMIXED is below:

```
/****run PROC NLMIXED to fit the general model ****/
proc nlmixed data=pkdata method=GAUSS;
parms theta1=20
      theta2=0.8
      theta3=0.6
      sigma_1=&sigma_1
      sigma_2=&sigma_2
      sigma_3=&sigma_3
      sigma=&sigma
      ;
v = theta1*exp(eta1);
ka = theta2*exp(eta2);
cl = theta3*exp(eta3);
sigma2_1=sigma_1**2;
sigma2_2=sigma_2**2;
sigma2_3=sigma_3**2;
sigma2=sigma**2;

k=cl/v;

cst=ka/(v*(ka-k));

pred = cst*dose_1*(exp(-k*time_1)-exp(-ka*time_1))
     +cst*dose_2*(exp(-k*time_2)-exp(-ka*time_2))
     +cst*dose_3*(exp(-k*time_3)-exp(-ka*time_3));

model y ~ normal(pred,sigma2);
random eta1 eta2 eta3 ~ normal([0,0,0],[sigma2_1,0,sigma2_2,0,0,sigma2_3])
    subject=id;
run;
```

In the example above, we presented the general case with constant dosage and nominal time points. This method can be easily modified to accommodate different situations. In the case of changing dosage during therapy, create the *dose_i* variables to indicate the correct dose amount for each dose. If actual time is used in the analysis instead of the nominal time, the variable *time_i* may be slightly different from the scheduled post-dose time, but the data structure and the PROC NLMIXED statements needed stay the same.

The general model can handle missed doses easily. For example, if a subject missed the 2nd dose and the 5th dose in a 7-day once-daily study, simply assign *dose_2* =0 and *dose_5*=0.

TO FIT THE SIMPLIFIED MODEL

As discussed in Section 2, for the special yet very common case with constant doses and nominal time, the simplified model (2) can be used for analysis, which can significantly reduce computational time.

To fit the simplified model (2), the required analysis data structure is much simpler than the one for the general model: variables *dose_i* and *time_i* for each dose are not needed. Using the same example, the table below illustrates the required data structure.

Table 3. Data Structure for Fitting the Simplified Model (2)

ID	n	time	y	dose	II
1	1	1	2.7878	100	24
1	1	4	4.8740	100	24
1	1	12	4.0114	100	24
1	1	24	3.3118	100	24
1	2	48	4.8601	100	24
1	3	49	7.5029	100	24
1	3	52	9.2563	100	24
1	3	60	7.8583	100	24
1	3	72	5.7894	100	24

The variable, II, is the inter-dose interval. In this example it is 24 hours, which means that dose is administrated every 24 hours.

Using the above data structure, the simplified model (2) is directly applied. The sample code below presents the mean function and the model statement in PROC NLMIXED.

```
cst=ka/(v*(ka-k));
pred = cst*dose*((1-exp(n*k*II))*exp(-k*time)/(1-exp(k*II))
                 -(1-exp(n*ka*II))*exp(-ka*time)/(1-exp(ka*II)));
model y ~ normal(pred,sigma2);
```

In the cases of missed dose(s), the simplified model (2) is not applicable, but its derived model (3) or (4) can be applied. The simplified model (3) is for the case with one missed dose, and the model (4) is for the case with multiple missed doses. It is important to note that, as the number of missed doses increases, the computational efficiency of (4) decreases with increased programming complexity, compared to the general model (1). Practically, one or two missed doses are not uncommon. In this paper, we only consider the case of one missed dose to illustrate the computational advantage of (3) and (4).

Here we assume that Subject 1 missed the 2nd dose in the above sample data. To use model (3), create a variable *dosem* to indicate the dose amount missed (see Table 4). In this case, *dosem* is 0 for all records before the scheduled time for the 2nd dose since the missed 2nd dose has no effect then. The variable *dosem* is 100 for all records after the scheduled time for the 2nd dose, since the missed dose has effect after this time point. If another subject missed a dose too (not necessary the 2nd dose), the variable *dosem* can be populated similarly: 0 for all records before the missed dose, and 100 for all records after the missed dose. For subjects who didn't miss a dose, *dosem* is 0.

Table 4. Data Structure for Fitting Simplified Model (3)

ID	n	time	y	dose	dosem	II
1	1	1	2.7878	100	0	24
1	1	4	4.8740	100	0	24
1	1	12	4.0114	100	0	24
1	1	24	3.3118	100	0	24
1	2	48	4.8601	100	100	24
1	3	49	7.5029	100	100	24
1	3	52	9.2563	100	100	24
1	3	60	7.8583	100	100	24
1	3	72	5.7894	100	100	24

The simplified model (3) contains two terms: the first term is the same as the model (2), and the second term accounts for the missed dose. For the subjects who didn't miss a dose, the mean function is actually the first term of the model (3). For the subjects who had a missed dose, the mean function is the first term of the model (3) for the records before the missed dose, and the mean function is exactly the same as the model (3) for the records after the missed dose. But with the data structure specified above, the model (3) can be applied to all records because *dosem* is 0 for the records without missed dose effect and then the second term is 0. To fit the simplified model (3), the statements for the mean function and the model statement are the following:

```
cst=ka/(v*(ka-k));
pred = cst*dose1*((1-exp(n*k*II))*exp(-k*time)/(1-exp(k*II))
                 -(1-exp(n*ka*II))*exp(-ka*time)/(1-exp(ka*II)))
      -cst*dosem*(exp(-k*(time-II)) - exp(-ka*(time-II)));
model y ~ normal(pred,sigma2);
```

4. CONCLUSIONS

In this paper, we illustrated a SAS template program for manipulating data and then constructed mean functions for fitting one-compartment models with PROC NLMIXED to pharmacokinetic data from multiple oral dose studies. Compared to the special case with constant dosage and regular dosing time, the required data structure for the general cases is more complex and the computation time can be very slow, but this more complex data structure may be necessary depending on the dosing schedule.

Although only one-compartment models for oral dose administration were discussed in the paper, the data structure and sample codes can be easily adapted for models with alternative route of administration and more complex structural and stochastic components.

REFERENCE

Beal, S.L. and Sheiner, L.B. (1998), *NONMEM User's Guides, Version V*, University of California, San Francisco.

Davidian, M. and Giltinan, D. (1995), *Nonlinear Models for Repeated Measurement Data,* New York, NY: Chapman and Hall.

ACKNOWLEDGMENTS

We are grateful to Amy Gillespie and Debbie Panebianco for their careful review and valuable comments.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the authors at

Jing Su
Merck & Co., Inc.
UG1CD-14
Po Box 1000
North Wales, PA 19454-1099
Email: jing_su@merck.com

Peng Sun
Merck & Co., Inc.
UG1D-44
Po Box 1000
North Wales, PA 19454-1099
Email: peng_sun@merck.com

Xiaodong Li
Merck & Co., Inc.
UG1D-44
Po Box 1000
North Wales, PA 19454-1099
Email: xiaodong_li@merck.com

Alan Hartford
Merck & Co., Inc.
UG1D-44
Po Box 1000
North Wales, PA 19454-1099
Email: alan_hartford@merck.com